TRANSFER LINES - TOOLS

Wolfgang Bartmann

CAS, Erice, March 2017

Wrap up

- Optics in a ring is defined by ring elements and **periodicity** – optics in a transfer line is dependent line elements and **initial conditions**



- **Changes** of the strength of a transfer line magnet **affect only downstream optics**



Wrap up

- Geometry calculations require a set of coordinates in a common reference frame

- **Bending fields** are defined by geometry and the **magnetic or electric rigidity**:

$$B\rho\ [Tm] = 3.3356\ \tfrac{A}{n}\ p\ [GeV/c] \qquad\qquad E\rho\ [kV] = \tfrac{\gamma+1}{\gamma}\ \tfrac{A}{n}\ T\ [keV]$$

$$\theta = \frac{Bdl}{B\rho}\ \text{or}\ \frac{Edl}{E\rho}$$

- The choice between magnetic and electric depends mainly on the beam energy

- If you are in the grey zone, consider: field design and measurement, power consumption, vacuum, interlocking

- For the estimates of bending radii in lines remember to take into account the filling factor (~70%) and Lorentz-Stripping in case of H^- ions

Wrap up

- **Quadrupole gradients and apertures** can be estimated in case of simple focussing structure like FODO cells

$$\frac{L}{f} = 4\sin\frac{\mu}{2}$$

Stability

$$f > \frac{L}{4}$$



$$\beta = \left(L + \frac{L^2}{4f}\right) / \sin\mu$$

Defines beam size and quadrupole pole tip field

- **Aperture specifications** require safety factors for the optics and constant contributions for trajectory variations and alignment errors
  - $A_{x,y} = \pm n_{sig} \cdot \sqrt{k_\beta \cdot \beta_{x,y} \cdot \frac{\epsilon_{x,y}}{\beta\gamma}} \pm D_{x,y} \cdot k_\beta \cdot \frac{\Delta p}{p} \pm CO \cdot \sqrt{\frac{\beta_{x,y}}{\beta_{xmax,ymax}}}$ $\pm$ alignment

Wrap up

- **Estimating tolerances** from emittance growth:

- Dipole field and alignment:

$$\epsilon_2 = \epsilon_1 + \frac{1}{2}\left((\Delta y)^2 \frac{1+\alpha^2}{\beta} + (\Delta y')^2 \beta\right)$$

Magnet misalignment

Dipole field error

$$\boxed{\Delta y' = \frac{\Delta B l}{B\rho}}$$

- Gradient errors:

$$\epsilon_2 = \frac{1}{2}\left(k^2\beta^2 + 2\right)\epsilon_1 \qquad k = -\frac{\Delta G l}{B\rho}$$

Outline

- Introduction
  - What is a transfer line?
- Paper studies
  - Geometry – estimate of bend angles
  - Optics – estimate of quadrupole gradients and apertures
  - Error estimates and tolerances on fields

1st hour

- Examples of using MADX for
  - Optics and survey matching
  - Final focus matching
  - Achromats
  - Error and correction studies

2nd hour

- Special cases in transfer lines
  - Particle tracking
  - Plane exchange
  - Tilt on a slope
  - Dilution
  - Stray fields

3rd hour

MADX example – Optics and survey matching

SPL to PS2 transfer line







SPL to PS2 optics

- Periodic FODO structure in the main part of the line
- Matching section from SPL to FODO and from FODO to PS2 injection
- Horizontal and vertical achromats in the 90 deg FODO – where they come for free
- Why do we provide dispersion at injection?



Christoph Hessler

```
TITLE, 'TTL1';
// read in sequence file
OPTION, -ECHO, WARN, -INFO;
CALL,    FILE = "ttl1-v1.0.0.seq";
// read in beam properties
call, file = "beam.madx";

//----------------------- Survey matching to PS2 -------------------------
USE,     SEQUENCE = TTL1;
// read in survey coordinates at the injection point (stripping foil)
// in the PS2 (real beam direction)
call, file = 'out.ps2foil-v8.0.4.survey.ini.mad';
value,  ps2.omk.foil.x0,
    ps2.omk.foil.y0,
    ps2.omk.foil.z0,
    ps2.omk.foil.theta0,
    ps2.omk.foil.phi0,
    ps2.omk.foil.psi0;

    // define macro for position matching routine:
m1: MACRO={USE, SEQUENCE = TTL1;
           SURVEY,   x0 = -1510.09247,
            y0 = 2423.58630,
            z0 = 2278.25946,
            theta0 = 0.8715909,
            phi0 = -0.0150853,
            psi0 = 0.0;};

// match position and angles at injection point (stripping foil):
MATCH, USE_MACRO;

//VARY, name = MB.v1.angle, STEP = 0.001,   LOWER = 0.01,   UPPER = 0.03;
VARY, name = MB.h1.angle,   STEP = 0.001,   LOWER =-0.06,   UPPER =-0.01;
VARY, name = MB.h2.angle,   STEP = 0.001,   LOWER = 0.01,   UPPER = 0.06;
//VARY, name = MB.v2.angle, STEP = 0.001,   LOWER = -0.033,UPPER =-0.01;
VARY, name = ps2.omk.foil.at,   STEP = 0.001,   LOWER = 400, UPPER = 425;

USE_MACRO, name= m1;

CONSTRAINT, expr= table(survey,ps2.omk.foil,x) = ps2.omk.foil.x0;
//CONSTRAINT,   expr= table(survey,ps2.omk.foil,y)=ps2.omk.foil.y0;
CONSTRAINT, expr= table(survey,ps2.omk.foil,z)=ps2.omk.foil.z0;
CONSTRAINT, expr= table(survey,ps2.omk.foil,theta)=ps2.omk.foil.theta0;
//CONSTRAINT,   expr= table(survey,ps2.omk.foil,phi)=ps2.omk.foil.phi0;
//CONSTRAINT,   expr= table(survey,ps2.omk.foil,psi)=ps2.omk.foil.psi0;

LMDIF, CALLS:=5000, TOLERANCE:=1.E-10;
//JACOBIAN,  CALLS= 500, TOLERANCE= 1.0E-21;
ENDMATCH;
```

Survey

- Call sequence and define BEAM command
- Get the survey coordinates x,y,z, theta, phi, psi of the stripping foil
- Define macro for matching routine

When matching the geometry of the line, the path length is changing → the position of the foil in the sequence has to be matched simultaneously

FODO

```
// FODO cell properties
l.fodo     = 25;
phadv      = pi/2.0;// 90deg phase advance per cell
k1.theo    = 4.0 / (l.mq * l.fodo) * sin(phadv/2.0);
value, l.fodo, phadv, k1.theo;

kqif.cell = k1.theo;//  5.98643e-02
kqid.cell =-k1.theo;// -5.98462e-02

//----------------------- Prototype Cell --------------------------------------

FCELL: SEQUENCE,    L = l.fodo;

MQF.b:MQH,  AT := l.MQ/4,                K1 := kqif.cell;
MQD.a:MQH,  AT := l.fodo*0.5-l.MQ/4,     K1 := kqid.cell;
MQD.b:MQH,  AT := l.fodo*0.5+l.MQ/4,     K1 := kqid.cell;
MQF.a:MQH,  AT := l.fodo-l.MQ/4,         K1 := kqif.cell;

ENDSEQUENCE;

USE,    SEQUENCE = FCELL;
MATCH,  SEQUENCE = FCELL;
    VARY,   NAME = kqif.cell, STEP = 0.001,   LOWER = 0.0, UPPER = mq.k1.max;
    VARY,   NAME = kqid.cell, STEP = 0.001,   LOWER = -mq.k1.max, UPPER = 0.0;
    CONSTRAINT, RANGE = #E, MUX = phadv/(2*pi), MUY = phadv/(2*pi);
LMDIF, CALLS:=50, TOLERANCE:=1.E-10;
ENDMATCH;

SELECT, FLAG=TWISS, COLUMN=NAME,S,BETX,BETY,ALFX,ALFY,DX,DPX,DY,DPY;
TWISS,  FILE = "fodo.twiss";
```

$$\frac{L}{f} = 4\sin\frac{\mu}{2}$$

Matching sections

```
//----------------------- Match TTL1 to SPL ----------
USE,     SEQUENCE = TTL1;
MATCH,   SEQUENCE = TTL1,    VLENGTH = TRUE,
                BETX    = BETX0,    ALFX    = ALFX0,
                DX   = DX0,      DPX = DPX0,
                BETY    = BETY0,    ALFY    = ALFY0,
                DY   = DY0,      DPY = DPY0;
    VARY,   NAME = kqid.0, STEP = 0.001,    LOWER = -0.1, UPPER = 0.0;
    VARY,   NAME = kqif.1, STEP = 0.001,    LOWER = 0.0, UPPER = 0.1;
    VARY,   NAME = kqid.1, STEP = 0.001,   LOWER = -0.1, UPPER = 0.0;
    VARY,   NAME = kqif.2, STEP = 0.001,    LOWER = 0.0, UPPER = 0.1;
    CONSTRAINT, RANGE = MQF.4,  BETX = MQF.betx, BETY = MQF.bety;
    CONSTRAINT, RANGE = MQF.4,  ALFX = MQF.alfx, ALFY = MQF.alfy;
LMDIF, CALLS:=5000, TOLERANCE:=1.E-10;
ENDMATCH;
//----------------------- Match TTL1 to PS2 --------------------------------
USE,     SEQUENCE = TTL1;
MATCH,   SEQUENCE = TTL1,    VLENGTH = TRUE,
                BETX    = BETX0,    ALFX    = ALFX0,
                DX   = DX0,      DPX = DPX0,
                BETY    = BETY0,    ALFY    = ALFY0,
                DY   = DY0,      DPY = DPY0;
    VARY,   NAME = kqif.13, STEP = 0.001,   LOWER = 0.04, UPPER = 0.1;
    VARY,   NAME = kqid.13, STEP = 0.001,   LOWER = -0.1, UPPER = 0.0;
    VARY,   NAME = kqif.14, STEP = 0.001,   LOWER = 0.04, UPPER = 0.071;
    VARY,   NAME = kqid.14, STEP = 0.001,   LOWER = -0.1, UPPER = 0.0;
    VARY,   NAME = kqif.15, STEP = 0.001,   LOWER = 0.04, UPPER = 0.07;
    VARY,   NAME = kqid.15, STEP = 0.001,   LOWER = -0.1, UPPER = 0.0;
    VARY,   NAME = kqif.16, STEP = 0.001,   LOWER =  0.0, UPPER = 0.1;
    VARY,   NAME = kqid.16, STEP = 0.001,   LOWER = -0.1, UPPER = 0.0;
    CONSTRAINT, RANGE = ps2.omk.foil,   BETX=BETX1, BETY=BETY1;
    CONSTRAINT, RANGE = ps2.omk.foil,   ALFX=ALFX1, ALFY=ALFY1;
    CONSTRAINT, RANGE = ps2.omk.foil,   DX=DX1,     DPX=DPX1;
    CONSTRAINT, RANGE = ps2.omk.foil,   DY=DY1,     DPY=DPY1;

LMDIF, CALLS:=5000, TOLERANCE:=1.E-10;
ENDMATCH;
```

SPL to FODO matching

FODO to PS2 matching

- Matching itself rather straightforward as long as you have enough knobs - independently powered quadrupoles
- Start generous to find optimum solution for optics
- Then minimize number of independent quadrupoles for economy – maybe even at the expense of non-perfect optics
- Tunability!

- Not so straightforward to get the optics constraints right
- Make a clear handover point – better something mechanical than magnetic
- Calculation of optics at handover point
  - Take into account kicker – will alter dispersion
  - Switch on injection bump

Example – Final focus matching

HiRadMat

- Material test facility at the SPS
- 440 GeV
- 5e13 protons per shot
- Adjustable focal point
- Beam size 0.1 – 2 mm



Doublet



$$\begin{pmatrix} 1 & 0 \\ -1/f_2 & 1 \end{pmatrix} \begin{pmatrix} 1 & d \\ 0 & 1 \end{pmatrix} \begin{pmatrix} 1 & 0 \\ 1/f_1 & 1 \end{pmatrix} = \begin{pmatrix} 1 + d/f_1 & d \\ 1/f_1 - 1/f_2 - d/f_1 f_2 & 1 - d/f_2 \end{pmatrix}$$

For equal quadrupole strength:

$$R = \begin{pmatrix} 1 + d/f & d \\ -d/f^2 & 1 - d/f \end{pmatrix} \Rightarrow f^* = \frac{f^2}{d}$$

Effective focal length

Point to parallel focusing

- Drift of length l preceding the doublet



- R_{22} and R_{44} need to be zero which gives

$$f_1 f_2 = ld \qquad \frac{f_1}{f_2} = \frac{l}{l+d}$$

- And for the magnetic fields:

$$f_1 = l\sqrt{\frac{d}{l+d}} \qquad f_2 = \sqrt{d(l+d)}$$

Symmetric point to point focusing - Triplet

- Use two doublets to focus point to point
- Also works with one doublet, but asymmetric in x/y
- Triplet – symmetric in x/y → can create round beam spots at target



$$R = \begin{pmatrix} 1 - 2\frac{d^2}{f^2} & 2d\left(1 - \frac{d}{f}\right) \\ -2\frac{d}{f^2}\left(1 + \frac{d}{f}\right) & 1 - 2\frac{d^2}{f^2} \end{pmatrix} \qquad f^* = \frac{f^2}{2d}\left(1 + \frac{d}{f}\right)$$

Transfer matrix and effective focal length of triplet structure

Doublet optics

- Compared to FODO, the doublet provides longer drift spaces for special equipment
- Steep asymmetric slopes in betatron functions



Triplet optics

- Triplets have distinct locations of high betatron functions but then allow to focus them down over long distances
- $\mu = \int \frac{ds}{\beta}$ – large phase advance for closing dispersion bumps
- Low betas for economic dipole production



HiRadMat



Triplet structure as final focus of HiRadMat to control both planes and different focal length

In this case focus down to minimum beam size in both planes – large peak of betatron function upstream

CNGS and AWAKE

CNGS



Final focus structure

AWAKE



AWAKE final focus





$\sigma_{x,y}$ = 200 um

- Displace existing magnets of final focusing to fulfill optics requirements at the entrance of the plasma cell
- Move existing dipole + 4 additional dipoles to create a chicane for laser mirror integration

C. Bracco, J. Schmidt

Achromat



$$R_{bend} = \begin{pmatrix} 1 & 0 & 0 \\ -hs & 1 & s \\ 0 & 0 & 1 \end{pmatrix}$$

$$h = \frac{B}{B\rho}$$

$$s = \sin\alpha$$

$$M = \begin{pmatrix} 1 & 0 & 0 \\ -hs_2 & 1 & s_2 \\ 0 & 0 & 1 \end{pmatrix} \begin{pmatrix} 1 & L_2 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix} \begin{pmatrix} 1 & 0 & 0 \\ -1/f & 1 & s_2 \\ 0 & 0 & 1 \end{pmatrix} \begin{pmatrix} 1 & L_1 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix} \begin{pmatrix} 1 & 0 & 0 \\ -hs_1 & 1 & s_1 \\ 0 & 0 & 1 \end{pmatrix}$$

[6] D. Carey. The Optics of Charged Particle Beams. 1987.

Achromat

- Multiply the matrix and determine the expressions for Dispersion and its derivative:

$$D_x = L_2 s_1 \left(1 - \frac{L1}{f}\right) + L_1 s_1$$

$$D'_x = L_1 s_1 \left(-h s_2 - \frac{1}{f} + \frac{h s_2 L_2}{f}\right) + s_1 \left(1 - h s_2 L_2\right) + s_2$$

- For an achromat, both must vanish which defines the strengths and lengths parameters in our system:

$$\frac{1}{f} = \frac{1}{L_1} + \frac{1}{L_2}$$

$$L_1 s_1 = L_2 s_2$$

Achromat with point to point focusing



- Central quadrupole is focusing the dispersion down to zero at the outer ends of the bends and contributes little to the overall focusing
- Quadrupoles outside the achromat do not affect the dispersion behaviour; they solely provide focusing in both planes
- Target at the centre can be used as spectrometer

Achromat examples from high brightness machines



P. Bryant.

Double bend achromat – Chasman-Green

Expanded DBA version

Strong focussing in dispersive area – consider installing sextupoles for chromaticity correction

Triple bend achromat

Resonant dispersion condition in arcs

- More relevant for rings
- Resonant condition for dispersion cancellation in an arc



[8] Senichev Yurij. A "resonant" lattice for a synchrotron with a low or negative momentum compaction factor. 1997.

MADX example – Error and correction studies

Error studies

- Define acceptable error levels of your line
  - Loss level – W/m, % beam, activation level…
  - Trajectory offset in position and angle at point of delivery
  - Shot-to-shot variation of trajectory
  - Long term variation of trajectory
  - Optics mismatch at point of delivery
  - Rotation, energy mismatch
- Assume reasonable errors for the hardware and study their impact
- Iterate input until error level is acceptable and specify hardware accordingly

MADX example - Error studies

- Input error from preceding machine
- According to optics at handover




```
corrmacro(nx): macro={
//use, sequence=btbtp4;
use, sequence=bt1btp;
sigma:=2;
vert:=TGAUSS(2);

EOPTION,SEED=nx,ADD=false;

eps_x=2e-6/2.9676;
//delta_p=1.07e-3;
delta_p=0;

rand_x=gauss(nx);
sigma_x = sqrt(eps_x * betx0);
x0 = 0.015*rand_x *( sigma_x + (abs(dx0) * delta_p));

rand_px=gauss(nx);
px0 = 0.015*rand_px *((sqrt(eps_x / betx0) - (alfx0 / betx0) * sigma_x)) + (dpx0 * delta_p);

eps_y=2e-6/2.9676;
//delta_p=1.07e-3;

rand_y=gauss(nx);
sigma_y = sqrt(eps_y * bety0);
y0 = 0.01*rand_y * (sigma_y + (abs(dy0) * delta_p));

rand_py=gauss(nx);
py0 = 0.01*rand_py *((sqrt(eps_y / bety0) - (alfy0 / bety0) * sigma_y)) + (dpy0 * delta_p);
```



Error assignment

```
EOPTION,ADD= FALSE, SEED:=nx;
SELECT, flag= error, clear;

SELECT, FLAG= ERROR, PATTERN= "BT.BHZ10";
EFCOMP, ORDER:= 0, Radius:=0.01,
DKNR:={5e-4*TGAUSS(sigma),0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0},
DKSR:={0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0};
EALIGN,  DS:= 0.3e-3*TGAUSS(sigma), DX:= 0.3e-3*TGAUSS(sigma), DY:= 0.3e-3*TGAUSS(sigma),
DPHI:= 0.3e-3*TGAUSS(sigma), DTHETA:= 0.3e-3*TGAUSS(sigma), DPSI:= 0.3e-3*TGAUSS(sigma);
```

What do these numbers for field error and reference radius mean for a magnet?

Multipole expansion of static magnetic field

The magnetic field in the aperture of the accelerator magnet is usually expanded as:

$$B_y + iB_x = B_{ref} \sum_{n=1} (b_n + ia_n) \left(\frac{x+iy}{R_{ref}^{n-1}} \right)^{n-1}$$

where bn, an, are normal and skew multipole coefficients. Rref is the reference radius, and Bref an appropriately chosen normalization.

For a long magnet, the vector potential A in the magnet aperture has only the longitudinal component A(0, 0, Az), with:

$$A_z = -B_{ref} \sum_{n=1} (b_n + ia_n) \frac{\mathrm{Re}(x+iy)^n}{nR_{ref}^{n-1}}$$

The Hamiltonian then directly contains the multipole coefficients.

$$H(x,y,z,t) = q\phi + \sqrt{\left(\frac{p_z - q\mathbf{A}}{1+k_x x} \right)^2 + p_x^2 + p_y^2 + m^2 c^4}$$

Main properties of multipole expansion

Multipole expansion:

- is analytical and satisfies automatically the **equations for the static field** in 2D.
- is based on a **complete and orthogonal** set of basis functions:
  - the value of the low order multipoles does not depend on the number of terms included in the expansion (important for precision of dipole, quadrupole and sextupole fields – stability of optical functions – which do not change if the series contains 8 or 10 terms, for example)
- guarantees convergence for all $r < R_{ref}$.
- on a circle is well matched to the rotating coil technique (low measurement and low data treatment errors)
- Most (if not all) optics codes use multipole expansion.
- If the aperture is very asymmetric (classical dipoles), multipole expansion can be performed on an elliptical boundary. Circular and elliptical multipoles are related by a linear transformation.

“Good Field Region”

- Concept related to iron-dominated dipole magnets, with an implicit assumption that the field deviations are largest at the borders of the region (near the pole).

- How to use the field plots?

- A field given in layers of y=const invites a fit with a polynomial in x. But:
  - Polynomials are not orthogonal
  - Do not necessarily satisfy the field equations.





Magnet errors in transfer lines

- The errors of the magnetic elements to be considered in the transfer lines are:
  - Linear terms: dipole and quadrupole strengths, including PC errors, transverse misalignments and rotations
  - First non-linear term: normal sextupole

  These errors affect the emittance growth, rms orbit errors, and chromatic properties of the transfer lines.
- The relevant factors are the strengths of magnetic field multipoles up to n=3 (sextupole).
- The field in the aperture, measured or calculated, should be obtained on the largest possible R_{ref}.
- The specification of the field should be given in terms of multipoles on a circle (or ellipse, in special cases).

Apply error distributions onto the beam



Trajectory correction

- Use small independently powered dipole magnets (correctors) to steer the beam
- Measure the response using monitors (pick-ups) downstream of the corrector (pi/2, 3pi/2, …)



- Horizontal and vertical elements are separated
- H-correctors and pick-ups located at F-quadrupoles (large betx)
- V-correctors and pick-ups located at D-quadrupoles (large bety)

π- bumps



Correction with some monitors disabled

With poor BPM phase sampling the correction algorithm produces a trajectory with 185mm y_{max}



Sufficient instrumentation is essential for trajectory correction

Trajectory correction

- Global correction can be used which attempts to minimise the RMS offsets at the BPMs, using all or some of the available corrector magnets.
- Steering in matching sections, extraction and injection region requires particular care
  - D and beta functions can be large $\rightarrow$ bigger beam size
  - Often very limited in aperture
  - Injection offsets can be detrimental for performance
  - Losses at collimators



Correction

```
SELECT,FLAG=ERROR,FULL;
ESAVE,FILE='errors.tfs';

select, flag=twiss, clear;
select, flag= twiss, column= s,x,y,betx,bety,px,py;
TWISS,file="outcorr/test/twissnocorr.nx.tfs", deltap= 0.0, sequence= bt1btp,BETA0=INITBETA0;

option,echo;
COPTION, PRINT=10;

CORRECT, flag = line, PLANE= x, MODE=svd, COND=0, MONON=1,  MONERROR=1, MONSCALE=0, RESOUT=0,
    ERROR=1.E-6, CORRLIM=1.0, CLIST="/orbit_correction/xcorr.nx.out";
CORRECT, flag = line, PLANE= y, MODE=svd, COND=0, MONON=1,  MONERROR=1, MONSCALE=0, RESOUT=0,
    ERROR=1.E-6, CORRLIM=1.0, CLIST="/orbit_correction/ycorr.nx.out";

select, flag= corr, column= PX.OLD,PY.OLD,PX.CORRECTION,PY.CORRECTION;

select, flag=twiss, clear;
select, flag= twiss, column= s,x,y,betx,bety,px,py;
TWISS,file="outcorr/test/twisscorr.nx.tfs", deltap= 0.0, sequence= bt1btp,BETA0=INITBETA0;
};

n=0;
while (n < 200) {

  exec, corrmacro($n);
 n= n+1;

}
```

Check that corrector kicks are not adding up
→ set them to 0 each time

Output

- Display uncorrected/corrected trajectories



- Compare to what is included in aperture definition (losses)



$$A_{x,y} = \pm n_{sig} \cdot \sqrt{k_\beta \cdot \beta_{x,y} \cdot \frac{\epsilon_{x,y}}{\beta\gamma}} \pm D_{x,y} \cdot k_\beta \cdot \frac{\Delta p}{p} \pm CO \cdot \sqrt{\frac{\beta_{x,y}}{\beta_{xmax,ymax}}} \pm \text{align}$$

Output

- Display uncorrected/corrected trajectories
- Compare to what is included in aperture definition (losses)
- Statistics on error at handover to ring – filamentation will occur
- Useful to express error in x, px, y, py in displacement vector form in normalized phase space – can directly estimate the expected emittance growth

$$\epsilon_2 = \epsilon_1 + \frac{1}{2}D^2$$





Output

- Display uncorrected/corrected trajectories
- Compare to what is included in aperture definition (losses)
- Statistics on error at handover to ring – filamentation will occur
- Useful to express error in x, px, y, py in displacement vector form in normalized phase space – can directly estimate the expected emittance growth
- Sensitivity analysis – which error is the most important one
  - Instrumentation sensitivity – reading/scaling error and full failure – pi bumps!

	Tolerance ΔI/I	x rms mm	p_x rms μrad	R_x^2/ϵ_0 1×10^{-3}	y rms mm	p_y rms μrad	R_y^2/ϵ_0 1×10^{-3}
Random effects							
PSB orbit ± 0.15/0.10 mm (h/v)		0.04	4	**0.4**	0.04	2	**0.2**
BVT10	1×10^{-4}				0.08	1	**0.3**
SMV10	1×10^{-4}				0.13	1	**1**
QNO10	5×10^{-4}				0.11	1	**1**
QNO20	5×10^{-4}				0.03	1	**0.06**
KFA10	3×10^{-4}				0.02	1	**0.06**
SMV20	1×10^{-4}				0.01	4	**1**
KFA20	3×10^{-4}				0.01	0	**0.02**
BVT20	1×10^{-4}				0.05	3	**1**
BT.BHZ10	1×10^{-4}	0.07	0.02	**4**			
All random effects		**0.08**	**17**	**5.1**	**0.21**	**6**	**4.0**
Systematic effects							
KFA10	5×10^{-3}				0.39	15	**17**
KFA20	5×10^{-3}				0.22	8	**5**

Specify transverse feedback

$$\epsilon_2 = \epsilon_1 + \frac{1}{2}\left((\Delta y)^2 \frac{1+\alpha^2}{\beta} + (\Delta y')^2 \beta\right)\left(\frac{1}{1+\tau_D C/\tau_d}\right)^2$$

Damping effect of transverse feedback

Need to specify the peak oscillation amplitude and bandwidth of the system







Vincenzo Forte

Typical specifications from correction studies

- Number of monitors and required resolution
  - Every ¼ betatron wavelength
  - Grid resolution: ~3 wires/sigma
- Number of correctors and strength
  - Every ½ betatron wavelength H - same for V
  - Displace beam by few betatron sigma per cell
- Dipole and quadrupole field errors
  - Integral main field known to better than 1-10 E-4
  - Higher order field errors < 1-10 E-4 of the main field
- Dynamic errors from power converter stability
  - 1-10e-5
- Alignment tolerances
  - 0.1-0.5 mm
  - 0.1-0.5 mrad

Take typical values as good guess starting point and refine them according to your simulations

Error source	tolerance $\Delta I/I_{nom}$	$\Delta\sigma_x$	$\Delta\sigma_y$
Random effects			
SPS Orbit $\pm$ 0.10mm		0.113	0.113
Line stability $\pm$ 0.20mm		0.226	0.226
MSE	$\pm$ 1.3E-04	0.104	0.000
BH1	$\pm$ 5.0E-05	0.083	0.000
MSI	$\pm$ 5.0E-05	0.096	0.000
MBI	$\pm$ 2.5E-05	0.171	0.000
BV1	$\pm$ 2.5E-05	0.000	0.112
rms sum (1 σ)		0.342	0.279
Systematic effects			
MKE (systematic)	$\pm$ 5.0E-03	0.224	0.000
MKI (systematic)	$\pm$ 5.0E-03	0.000	0.354

Summary

- Before switching on a computer we can define for a transfer line
  - Number of dipoles and quadrupoles, correctors and monitors
  - Dipole field and quadrupole pole tip field
  - Aperture of magnets and beam instrumentation
  - Rough estimate of required field quality and alignment accuracy

- With computer codes
  - We can calculate the optics for matching sections and final focus for fixed target beams
    - Give a precise value for the field in each dipole and quadrupole
  - We can run error and correction studies
    - Define misalignment tolerances
    - Define field homogeneity and ripple
    - Define sensitivity of instrumentation
    - Define specifications for transverse feedback systems

Wrap up - optics

$$\frac{L}{f} = 4\sin\frac{\mu}{2}$$

- FODO
  - Classical choice for high flux transport
  - Analytically straightforward – can create achromat ‘by eye’
  - Good aperture and correction behaviour



$$f^* = \frac{f^2}{d}$$

- Doublet
  - Provides space for equipment
  - Steep, asymmetric in beta functions
  - Can focus from point source to parallel beams and point-to-point (secondary beam lines)



$$f^* = \frac{f^2}{2d}\left(1 + \frac{d}{f}\right)$$

- Triplet
  - Locally very high beta functions
  - Can focus beta to low values over long distance – good for dipole aperture
  - Can provide large phase advance over short distance – good to close a dispersion bump, achromat
  - Versatile for final focus matching and secondary beam lines (point to parallel – parallel to point)



Wrap up - achromat

- Straight forward dipole locations in a FODO lattice





- Spectrometer functionality if

$$\frac{1}{f} = \frac{1}{L_1} + \frac{1}{L_2}$$

$$L_1 s_1 = L_2 s_2$$



Wrap up – error/correction studies

- Make sure magnet and optics designer speak about the same errors
- Multipole expansion is a useful language
- For a transfer line errors up to order n=3 (sextupole) are relevant
- Trajectory correction in a line is straight forward
  - Specify sensitivity of instrumentation – be aware of pi-bumps in case of failure
- Apply errors systematically (distributed error function, sensitivity check) and evaluate effect on beam quality (losses, emittance)

Thank you for your attention

And many thanks to my colleagues for helpful input:

R. Baartman, D. Barna, M. Barnes, C. Bracco, P. Bryant, F. Burkart, V. Forte, M. Fraser, B. Goddard, C. Hessler, D. Johnson, V. Kain, T. Kramer, A. Lechner, J. Mertens, R. Ostojic, J. Schmidt, L. Stoel, C. Wiesner

[1] B. Goddard. Transfer lines. *CAS*, 2004.

[2] N. Ibarrola and M. Jones. Analysis and Evaluation of the CERN Reference Systems. *International Workshop of Accelerator Alignment*, 2016.

[3] M Fraser et al. Beam Dynamics Studies of the ELENA Electrostatic Transfer Lines. (CERN-ACC-2015-340):MOPJE044. 4 p, 2015.

[4] D. Johnson. Challenges Associated with 8 GeV H- Transport and Injection for FERMILAB PROJECT-X *. (Proceedings of Hadron Beam 2008), 2008.

[5] P. Bryant. Beam Transfer Lines. (CAS), 1992.

[6] D. Carey. The Optics of Charged Particle Beams. 1987.

[7] J. Rossbach and P. Schmueser. Basic Course on Accelerator Optics. (CAS), 1992.

[8] Senichev Yurij. A "resonant" lattice for a synchrotron with a low or negative momentum compaction factor. 1997.

[9] Werner Herr. Mathematical and Numerical Methods for Non Linear Beam Dynamics in Rings. CAS 2013.